DANIEL F. MERRIAM, Editor

# FORTRAN IV PROGRAM USING DOUBLE FOURIER SERIES FOR SURFACE FITTING OF IRREGULARLY SPACED DATA


By

**WILLIAM R. JAMES**
Northwestern University



**COMPUTER CONTRIBUTION 5**
State Geological Survey
The University of Kansas, Lawrence
1966

## Editor's Remarks

A Fourier series, a mathematical series of sines and cosines, can be used to analyze data suspected of being oscillatory. If the analysis indicates that the fit is poor, the data probably are not periodic; if the fit is good, then possibly they are. Inasmuch as a periodic function is predictable, the investigator is provided with a powerful analytical tool.

Natural cycles can be observed in sun-spot activity, tides, tree-ring growth, weather patterns, and many other phenomena. The geological record is more obscure, but many cyclic phenomena can be interpreted in sedimentary sequences. Cyclic sedimentation is now receiving more attention (see D. F. Merriam, ed., 1964, Symposium on Cyclic Sedimentation, Kansas Geological Survey Bull. 169), and results should be forthcoming from this important aspect of geology.

Recently, Fourier analysis has been successfully applied in geology to phenomena believed to be cyclic in nature. Analyses have been made of varved sequences (R. Y. Anderson and L. H. Koopmans, 1963, Harmonic analysis of varve time series, Jour. Geophys. Res., v. 68, no. 3), electric log representation of sediments (F. W. Preston and J. H. Henderson, 1964, Fourier series characterization of cyclic sediments for stratigraphic correlation, Kansas Geological Survey Bull .169), microrelief (R. O. Stone and J. Dugundji, 1965, A study of microrelief - its mapping, classification, and quantification by means of a Fourier analysis, Eng. Geol., v. 1, no. 2), and other mapped properties such as structural, geochemical, topographic, and stratigraphic data (see, reference list).

One of the problems in using Fourier analysis is the necessity of having orthogonally oriented data. In geology, of course, this criterion is seldom, if ever, met. This program, which will accept irregularly spaced data, will greatly facilitate the application of Fourier analysis in solving geologic problems.

For a limited time the Survey will make available the card deck for $10.00. Two versions of the program are available, one in FORTRAN IV for the CDC 3400, and the other in FORTRAN IV for the IBM 7040. Because the two versions are slightly different, it is necessary to specify which program is desired when ordering the deck.

The Kansas Geological Survey is the only geological organization known to be actively distributing computer program decks as well as data decks. The programs are sold for a limited time at a nominal cost. Versions of the programs have been executed on Burroughs B5500, CDC 3400, Elliott 803C, IBM 1620, 7040, 7090, and 7094/1401 computer systems. A list of available decks is given below.

| | ALGOL | FORTRAN II | FORTRAN IV |
|---|---|---|---|
| Marine Simulation (CC 1) | | | $20.00 |
| 2D Regression (CC 2) | $10.00 | $10.00 | $10.00 |
| Trend-6 (CC 3) | | | $25.00 |
| Discrim (CC 4) | | $ 5.00 | |
| Nongridded Double | | | |
| Fourier (CC 5) | | | $10.00 |
| *Trend-3 (SDP 3) | | | |
| Match-Coeff (SDP 4) | | $ 2.00 | |
| *Correlation and distance | | | |
| Coeff (SDP 9) | | | |
| Time-trend (SDP 12) | | $ 5.00 | $ 5.00 |
| Covap (SDP 13) | | $15.00 | |
| Trend-3 (SDP 14) | | $25.00 | $25.00 |
| Cross-Association (SDP 23) | | | $10.00 |
| Single and | | | $ 5.00 |
| double Fourier (SDP 24) | | | $15.00 |
| Precambrian wells (SDP 25) | | | |
| List of about 2,6000 Precambrian wells $50.00 | | | |
| Trend-4 (SDP 26) | | $ 7.50 | |
| Sediment analysis (SDP 28) | | $10.00 | |
| 4D Trend (KGS B171) | | $10.00 | $10.00 |
| Conversion of T&R to | | | |
| Cartesian coordinates (B 170-3) | | $ 5.00 | $ 5.00 |
| Hydrodynamic oil-trap mapping | | | |
| (reprint, Colo. Sch. Mines) | | | $10.00 |

*Out of print, therefore not available.

Comments and suggestions concerning the Computer Contribution Series are welcome and should be addressed to the editor. An up-to-date list of publications is available on request.

# FORTRAN IV PROGRAM USING DOUBLE FOURIER SERIES FOR SURFACE FITTING OF IRREGULARLY SPACED DATA

By

**WILLIAM R. JAMES**

## INTRODUCTION

The double Fourier series has recently come into use in geology as an alternative model to the polynomial for trend-surface analysis. Thus far its use has been restricted to gridded data due to the ease with which coefficients are computed. The extension of the model to irregularly spaced data greatly facilitates applications in geology.

Coefficients of the series are calculated by the least-squares method. Where data are arranged on a grid with fundamental wavelengths chosen as the grid lengths plus one, coefficients are easily obtained in great number. If the fundamental wavelengths are chosen as some other values or the data are irregularly distributed, computation of coefficients is more cumbersome but may be of greater advantage. Because there is essentially no restriction on choice of fundamental wavelengths in the latter method these values may be chosen on a substantive basis and hence may lead to more meaningful trend maps. Also, fundamental wavelengths may be chosen such that some extrapolation beyond the control area is possible, whereas in the gridded case the surfaces merely repeat themselves upon extrapolation beyond the control grid. This paper presents a review of the method of coefficient derivation and application to trend surfaces, a discussion of the differences between the gridded and nongridded applications of the method, a computer program for computation of coefficients and the mapping of surfaces, and examples of applications of the method.

Acknowledgments.--The writer is indebted to Dr. W. C. Krumbein for discussions leading to the development of this program and aid in constructing the map examples, and to Dr. John Tukey and the Statistical Techniques Research Group, at Princeton University for stimulating discussion regarding this subject.

## REVIEW OF THEORY AND APPLICATION OF FOURIER SERIES TO GRIDDED DATA

A finite definition of the double Fourier series useful for surface fitting to gridded data is given in equation (1). This brief review is based mainly on James (1966), and is included here for completeness.

$$F(U,V) = \sum_{i=0}^{KC} \sum_{j=0}^{LC} cc_{ij}\cos(2\pi iU/M)\cos(2\pi jV/N) + \sum_{i=0}^{KC} \sum_{j=1}^{LS} cs_{ij}\cos(2\pi iU/M)\sin(2\pi jV/N) + \sum_{i=1}^{KS} \sum_{j=0}^{LC} sc_{ij}\sin(2\pi iU/M)\cos(2\pi jV/N) + \sum_{i=1}^{KS} \sum_{j=1}^{LS} ss_{ij}\sin(2\pi iU/M)\sin(2\pi jV/N) \qquad (1)$$

where:

U = north-south coordinate axis increasing to the south with origin at northern map edge*
V = east-west axis increasing to the east with origin at the western map edge
$X_{(U,V)}$ = mapped variable
M = maximum U value plus one (fundamental wavelength in U direction)
N = maximum V value plus one (fundamental wavelength in V direction)
KC = maximum cosine harmonic in U direction
LC = maximum cosine harmonic in V direction
KS = maximum sine harmonic in U direction
LS = maximum sine harmonic in V direction
If M is even, KC = M/2; KS = (M-2)/2
If N is even, LC = N/2; LS = (N-2)/2
If M is odd, KC = KS = (M-1)/2
If N is odd, LC = LS = (N-1)/2

Coefficients of the series are $cc_{ij}$, $cs_{ij}$, $sc_{ij}$, and $ss_{ij}$.

*Notation of U,V,X in this paper corresponds to Y,X,Z of other papers published in the Computer Contribution Series.

The series is linear with respect to its coefficients and thus the least-squares method may be used to calculate the coefficients. In matrix notation the problem consists of the solution of the equation $S\hat{\beta} = g$, which is: $\hat{\beta} = S^{-1}g$ where $\hat{\beta}$ is the column vector of coefficients, S is the square matrix of sums of squares and cross-products of Fourier series terms, and g is a column vector of the sums of products of observed values and individual Fourier series terms. This matrix equation may be written as shown in Table 1.

If the definition of the double Fourier series given in equation (1) is used and if the data are on a grid, the S matrix is remarkably simple in that all elements outside the main diagonal are identically equal to zero and the elements of the main diagonal are simple functions of M and N. The inversion of the S matrix becomes insignificant as the inverted matrix has only a main diagonal, the elements of which are merely the reciprocals of the corresponding elements of S. Thus for gridded data and the given definition of the Fourier series, computation of the S matrix need not be made by computer. Coefficients may be calculated directly by the following formulas.

By using these formulas, as many as several thousand coefficients may be calculated in a few minutes with a high-speed digital computer. Computer programs available for such calculations have

$$cc_{ij} = (\omega/(MN)) \sum_{U=0}^{M-1} \sum_{V=0}^{N-1} X_{(U,V)} \cos(2\pi iU/M)\cos(2\pi jV/N)$$

$$cs_{ij} = (\omega/(MN)) \sum_{U=0}^{M-1} \sum_{V=0}^{N-1} X_{(U,V)} \cos(2\pi iU/M)\sin(2\pi jV/N)$$

$$sc_{ij} = (\omega/(MN)) \sum_{U=0}^{M-1} \sum_{V=0}^{N-1} X_{(U,V)} \sin(2\pi iU/M)\cos(2\pi jV/N)$$

$$ss_{ij} = (\omega/(MN)) \sum_{U=0}^{M-1} \sum_{V=0}^{N-1} X_{(U,V)} \sin(2\pi iU/M)\sin(2\pi jV/N) \quad (2)$$

where:

$\omega$ = 1 if i = 0 (or KC, when M is even) <u>and</u> j = 0 (or LC when N is even)

$\omega$ = 2 if i = 0 (or KC, when M is even) <u>or</u> j = 0 (or LC when N is even)

$\omega$ = 4 if i is not zero (or KC when M is even) <u>and</u> j is not zero (or LC when N is even)

been written by Harbaugh and Preston (1965), Preston and Harbaugh (1965), and James (1966).

Because the Fourier series is a continuous function, the terms may be combined in a variety of ways to produce trend-surface and residual contour maps, or to objectively contour the data itself (James, 1966; Krumbein, 1966). Trend surfaces may be constructed by grouping the coefficients according to their wavelengths. In Figure 1, the Fourier

Table 1.- Matrix equation for least-squares determination of Fourier series coefficients (after James, 1966).

$$
\begin{vmatrix}
\sum_{UV}(A_0C_0)^2 & \sum_{UV}A_1C_0A_0C_0 & \cdots\cdots \sum_{UV}B_{KS}D_{LS}A_0C_0 \\
\sum_{UV}A_0C_0A_1C_0 & \sum_{UV}(A_1C_0)^2 & \cdots\cdots \sum_{UV}B_{KS}D_{LS}A_1C_0 \\
\vdots & \vdots & \vdots \\
\sum_{UV}A_0C_0A_3D_1 & \sum_{UV}A_1C_0A_3D_1 & \cdots\cdots \sum_{UV}B_{KS}D_{LS}A_3D_1 \\
\vdots & \vdots & \vdots \\
\sum_{UV}A_0C_0B_{KS}D_{LS} & \sum_{UV}A_1C_0B_{KS}D_{LS} & \cdots\cdots \sum_{UV}(B_{KS}D_{LS})^2
\end{vmatrix}
\begin{vmatrix}
cc_{00} \\ cc_{10} \\ \vdots \\ cs_{31} \\ \vdots \\ ss_{KSLS}
\end{vmatrix}
=
\begin{vmatrix}
\sum_{UV}X_{(U,V)}A_0C_0 \\ \sum_{UV}X_{(U,V)}A_1C_0 \\ \vdots \\ \sum_{UV}X_{(U,V)}A_3D_1 \\ \vdots \\ \sum_{UV}X_{(U,V)}B_{KS}D_{LS}
\end{vmatrix}
$$

$$S \qquad \hat{\beta} \quad = \quad g$$

Notation:

$A_i = \cos(2\pi iU/M)$ $\qquad B_i = \sin(2\pi iU/M)$

$C_j = \cos(2\pi jV/N)$ $\qquad D_j = \sin(2\pi jV/N)$




Figure 1.- Diagram displaying grouping of double Fourier series coefficients according to wavelength (after James, 1966).

coefficients are arranged in a diagram to indicate this grouping. Block 0 contains only one term which produces a horizontal plane at the value of its coefficient. Block I contains eight terms which represent the fundamental wavelength surfaces. The sum of blocks 0 and I represents the first harmonic trend surface. Block II contains 16 terms which represent surfaces of one half the fundamental wavelengths. The sum of terms in blocks 0, I, and II represents the second harmonic surface. Each successive harmonic surface is constructed by adding the terms in the next block.

For the case of gridded data the coefficients are independent of the combination of Fourier terms used for a given surface. Thus a $Z^2$ array analogous to that developed for orthogonal polynomials may be built with computational ease. The $Z^2$ values can be used to separate trends from residuals in the data.

## RESTRICTIONS ON USE OF GRIDDED MODEL

Although advantages of the gridded model are great, there are some restrictions inherent in its use. The first restriction is that the fundamental wavelengths used in the series definition (M and N in equation 1) are determined only by the map dimensions and have no necessary relation to any wave forms which might actually be present in the data. This feature of the model could conceivably lead a careless investigator to interpretations of wave forms in his data which have been introduced entirely by the model. This arbitrary selection of wavelengths also excludes the possibility of extrapolation of trend surfaces. All surfaces will begin to duplicate themselves in the U and V directions beyond the control grid.

## EXTENSION OF MODEL TO IRREGULARLY SPACED DATA AND INCOMPLETE GRIDS

The matrix equation (equation 2) and the general application of this equation to the Fourier series shown in Table 1 are valid regardless of the distribution of data points or the selection of fundamental wavelengths. The only change in the procedure for deriving coefficients for the nongridded case is that the S matrix now contains nonzero elements outside the main diagonal and all the elements of this matrix must be computed and the matrix inverted. This reduces the number of coefficients which may be calculated in comparison to the case of gridded data. However, it enables one to choose nearly any fundamental wavelengths for the series that are desired.

These wavelengths may be selected by using autocorrelation techniques or by substantive judgment on a theoretical or observational basis. The only general restriction on these choices is that the fundamental wavelengths must be larger than the dimensions of the control area, because the surfaces duplicate themselves over intervals equal to the fundamental wavelengths in the U and V directions. By setting the fundamental wavelengths sufficiently larger than the dimensions of the control area, the abrupt repetition of trends outside the control area is avoided Thus extrapolation beyond the control area is feasible.

## PROGRAM DESCRIPTION

The basic purpose of this FORTRAN IV program is to compute the coefficients of selected double Fourier series terms by the least-squares method. The coefficients to be derived are selected by the investigator prior to computation and may be selected to isolate trend surfaces as discussed previously, or on any other basis desired. Three options are available. These options produce the following information.

OPTION 1.

a. Trend and residual values at each data point are calculated.
b. Data mean and variance and standard deviation are calculated.
c. The reduction in total corrected sum of squares of the data due to removal of the trend, and the F ratio calculated from variance estimates made from the trend and residual surfaces are computed.

OPTION 2. A continuous symbol (COSY) map of the trend surface is produced.

OPTION 3. This option provides a total Fourier series fit to the observed values, produces computed values and residuals for a check on machine rounding error, and produces COSY maps of the total fit surface and the trend residual surface. This option may not be taken if the number of data points exceeds an upper limit based on machine storage capacity. As the program is presently written, this upper limit is 90 data points, but this limit will be different for different machines and may be changed by altering a DIMENSION statement.

## PROGRAM INPUT

Several decisions must be made by the user before the program may be run. These decisions include the choice of the fundamental wavelengths in the U and V directions, the origin of the wave forms on the U, V grid, the double Fourier series terms to be included in the trend, and the double Fourier series terms to be used for the total fit if that option is taken.

The fundamental wavelengths (i.e., the values M and N in equation 1) may be chosen in any manner so long as they are larger than the control area. If the user suspects some periodicity in his data he may estimate its wavelength and multiply it by successive integers until a length greater than the control area is obtained. This length may then be used as the fundamental wavelength and terms of the suspected wavelength may be included in the trend. If the user chooses, he may run the program several times using a variety of fundamental wavelengths, choosing from them on the basis of the reduction in sum of squares due to removal of the trend.

The origin of the wave form may be any U, V, point on or off the control area. The choice of this origin does not affect the shape of the derived surfaces or the reduction in the sum of squares. It will, however, change the numerical value of the coefficients.

The choice of the double Fourier series terms included in the trend surface may be made in several ways. The most common way, as discussed previously is to group the terms according to their wavelengths. This has the effect of separating broad scale variability from local variability in a manner analogous to polynomial trend surfaces (Krumbein, 1966; James, 1966). In many cases other combinations of terms may be useful to define a trend. For example, an investigator may wish to isolate a suspected periodicity in his data and thus will wish to choose terms of the proper wavelength to include in the trend surface.

If the total fit option is taken, the double Fourier series must be expanded to include the same number of terms as there are data points. This expansion may be done (using equation 1) by stepping i and j alternately by one integer until the proper number of terms is obtained. In other words, the terms included are defined by expanding the box diagram in Figure 1 in the direction indicated by the arrows until a sufficient number of terms is included.

In order to define the various double Fourier series terms to be used in the computations, the following code is used.

i = the wave frequency in the U direction over the length of one fundamental wavelength (same as in equation 1).

j = the wave frequency in the V direction, etc.

T = code number indicating the type of double Fourier series term used.

T = 1 if cos U cos V is the type
T = 2 if cos U sin V is the type
T = 3 if sin U cos V is the type
T = 4 if sin U sin V is the type

With this code a set of three digits (ijT) uniquely defines a term in the double Fourier series.

After the above decisions have been made, the input deck is made as follows.

I. Four title cards (9A8)

For carriage control a "1" must be punched in the first column of the first card, and "0" 's in the first column of the other three cards. The titles are alphanumeric characters.

II. Master card (2F6.0, 3I1, x, 2I3, 2F6.0)

| col. | variable | description |
|---|---|---|
| 1-6 | UW | fundamental wavelength in U direction |
| 7-12 | VW | fundamental wavelength in V direction |
| 13 | OPA | option for computed trend values and residuals; "0" deletes these computations; "1" includes them |
| 14 | OPB | option for COSY map of trend surface; "0" deletes map; "1" includes map |
| 15 | OPC | option for complete fit computations (cannot be taken if number of data points exceeds ninety); "0" deletes computation of total fit; "1" includes computation of total fit |
| 16 | not used at present | |
| 17-19 | M | number of terms in trend |
| 20-22 | N | number of data points |
| 23-28 | UO | origin of wave form on U axis |
| 29-34 | VO | origin of wave form on V axis |

III. Format card (9A8)

Data card format. This format must include space for control point identification, U value, V value, and observed data value.

IV. Map control card (6F6.0)

| col. | variable | description |
|---|---|---|
| 1-6 | UL | minimum U value to appear on map |
| 7-12 | UH | maximum U value to appear on map |
| 13-18 | VL | minimum V value to appear on map |
| 19-24 | VH | maximum V value to appear on map |
| 25-30 | BASE | base of the values to be contoured |
| 31-36 | CINT | contour interval |

V. Trend surface identification card (26(3I1))

This card contains M (number of terms in trend) sets of three digits (i, j, T), each set identifying a term of the double Fourier trend. All digits are punched consecutively across the card.

VI. Data deck (Format given by format card)

Each data card includes an integer control point identifier, a U value, a V value, and the value of the dependent variable at that data point.

VII. Total fit double Fourier series term identification card (26(3I1)) Include only if option 3 is taken.

This card is set up the same way as the trend-surface identification card except it contains N (number of data points) sets of three digits (i, j, T). Up to four such cards may be necessary.

VIII. Next data deck, if any, beginning with the cards as above.

## ADDED NOTES ON PROGRAM

The mapping portion of the program is independent of the rest of the program. Because the double Fourier series is a continuous function, maps including any area within or without the control area may be called for by merely changing the values of UL, UH, VL, and VH on the map control card. The maps duplicate themselves over every interval equal to the fundamental wavelength and thus extrapolation beyond this interval usually lacks meaning. If however, the chosen fundamental wavelengths are much larger than the width of the control area, extrapolation of the trend surfaces may be of great value and this should be considered when specifying values for UL, UH, VL, and VH.

It is also a feature of this program that only 20 contour intervals are available for mapping. Values of residuals will fluctuate about the zero plane approximately within the range of $\pm$ 1 standard deviation of the data. Therefore if residual maps are to be considered, the base of the contour scale should be negative and numerically large enough in magnitude to reach all values. For example, if the data ranges from values of 1,000 to 3,000 with a mean of 2,000, the value for BASE should be -1,000 and CINT should be 200 if all the maps are to be studied. If the residual map is of no interest, the base may be set at any convenient value near the lower limit of the data range.

## SUGGESTED FUTURE ADDITIONS AND ALTERATIONS TO PROGRAM

1. An option may be added whereby iteration would be made using a variety of fundamental wavelengths giving the worker a wide choice of trends to be considered.

2. An option could be added such that maps would be produced over a range of U, V, limits such that recomputation of coefficients need not be made for maps for which extrapolation is desired.

3. The number of data points for which computation of total fit is possible is limited only by machine storage capacity. This program is written for a CDC 3400 computer. If total fit to more than ninety data points is desired, the program could be rewritten to exclude computation of trends and residuals and thus increase the number of data points for which total fit is possible. Similarly, a machine with larger storage capacity will allow total fit to a larger number of data points.

Listing of FORTRAN IV statements in Fourier program for irregularly spaced data.

```
      PROGRAM FOURFIT
C***************************************
C     A DOUBLE FOURIER SURFACE FITTING PROGRAM FOR
C     IRREGULARLY SPACED DATA
C     PROGRAM BY W. R. JAMES, NORTHWESTERN UNIVERSITY, JUNE, 1966
C***************************************
      DIMENSION TITL(36),FMT(9),IDEN(200),II(100),JJ(100),D(200),
     XG(100),IDT(100),SC(100),C(100),COMP(200),RES(200),U(200),V(200)
      COMMON/BB/ VV(70) ,UU(42),CH(20), BASE, CONTOUR(19), BLANK,
     XS(90,90),  P,UW,VW,UL,UH,VL,VH,UO,VO,CVAL(42,70)
C***************************************
C    READ CONTROLS AND DATA
C***************************************
    1 READ(60,100), TITL
      IF(EOF,60)2,3
    2 STOP
    3 PRINT 100, TITL
  100 FORMAT(9A8)
      ITOP =0
      READ(60,101) UW,VW,OPA,OPB,OPC,OPD,M,N,UO,VO
  101 FORMAT(2F6.0,4F1.0,2I3,2F6.0)
      PRINT 200,UW,VW,OPA,OPB,OPC,M,N,UO,VO
  200 FORMAT(///5X,*MASTER CARD*,5X,*U WAVELENGTH*,3X,F8.2,10X,*V WAVELE
     XNGTH*,3X,F8.2,//10X,*COMPUTED VALUES*,3X,F1.0,5X,*COSY MAP*,3X,
     XF1.0,5X,*COMPLETE FIT*,3X,F1.0,//10X,*NO COEFF IN TREND*,2X,I3,5X,
     X*NO DATA PTS*,3X,I6,5X,*WAVE ORIG*,2X,2F6.2)
      READ(60,100) FMT
      IF(OPB .EQ. 1) 4,5
    4 READ(60,102) UL,UH,VL,VH,BASE,CINT
      PRINT 4100,UL,UH,VL,VH,BASE,CINT
 4100 FORMAT(////10X,*MAP CARD*,5X,*UMIN*,2X,F6.2,5X,*UMAX*,2X,F6.2,
     X5X,*VMIN*,2X,F6.2,5X,*VMAX*,2X,F6.2,//20X,*BASE*,2X,F8.2,10X,
     X*CONTOUR INTERVAL*,2X,F6.2)
      DO 135 K=1,19
  135 CONTOUR(K)=BASE + (CINT*K)
    5 READ(60,103) (II(I),JJ(I),IDT(I),I=1,M)
  103 FORMAT(26(3I1))
      DO 6 K=1,N
      READ(60,FMT) IDEN(K),U(K),V(K),D(K)
    6 CONTINUE
  102 FORMAT(6F6.0)
  203 FORMAT(2X,4(I4,2X,F5.2,2X,F5.2,2X,F10.4,2X))
```

```
C****************************************
C  COMPUTE S, BETA HAT, AND G MATRICES
C****************************************
      P=2.0*3.1415926
   76 DO 10 I=1,M
      G(I)=0.0
      DO 10 J=1,M
   10 S(I,J)=0.0
      DO 11 K=1,N
      UA=(P*(U(K)-UO))/UW
      VA=(P*(V(K)-VO))/VW
      DO 12 I=1,M
      UII=II(I)*UA
      VII=JJ(I)*VA
      IF(IDT(I) .EQ. 1)16,13
   13 IF(IDT(I) .EQ. 2)17,14
   14 IF(IDT(I) .EQ. 3) 18,15
   15 SC(I)=SINF(UII)*SINF(VII)
      GO TO 19
   16 SC(I)=COSF(UII)*COSF(VII)
      GO TO 19
   17 SC(I)=COSF(UII)*SINF(VII)
      GO TO 19
   18 SC(I)=SINF(UII)*COSF(VII)
   19 G(I)=G(I)+SC(I)*D(K)
   12 CONTINUE
      DO 20 I=1,M
      DO 20 J=1,M
   20 S(I,J)=S(I,J)+ SC(I)*SC(J)
   11 CONTINUE
C****************************************
C      SOLVE FOR COEFFICIENTS
C****************************************
      K=0
   21 CONTINUE
      K=K+1
      IF(K .EQ. M)25,22
   22 DO 23 I=K,M
      IF(ABSF(S(I,K)) .LE. 0.0001) GO TO 3000
      XY=S(I,K)
      G(I)=G(I)/XY
      DO 95 J=K,M
   95 S(I,J)=S(I,J)/XY
 3000 CONTINUE
   23 CONTINUE
      L=K+1
      DO 24 I=L,M
      IF(S(I,K) .EQ. 1.0) 3001,3002
 3001 CONTINUE
      G(I)=G(I)-G(K)
      DO 24 J=K,M
      S(I,J)=S(I,J)-S(K,J)
 3002 CONTINUE
   24 CONTINUE
      GO TO 21
   25 C(M)=G(M)/S(M,M)
      L=M-1
      DO 30 I=1,L
   30 C(I)=G(I)
      I=M
   26 CONTINUE
      I=I-1
```

```
      K=I+1
      IF(K .EQ. 1)29,27
   27 DO 28 J=K,M
   28 C(I)=C(I)-S(I,J)*C(J)
      GO TO 26
   29 CONTINUE
      PRINT 1001
 1001 FORMAT(1H1//20X,*COEFFICIENTS*,//10X,*USUB*,4X,*VSUB*,4X,*TYPE*,
     X4X,*COEFFICIENTS*//)
      DO 180 K=1,M
      PRINT 1000,II(K),JJ(K),IDT(K),C(K)
  180 CONTINUE
 1000 FORMAT(10X,3(I4,4X),F12.6)
C*****************************************
C     COMPUTED VALUES AND RESIDUALS
C*****************************************
      IF(OPA .EQ. 1)31,50
   31 PRINT 110
  110 FORMAT(1H1,30X,*COMPUTED VALUES AND RESIDUALS*,//10X,
     X*IDEN*, 8X,*U*,9X,*V*,8X,*OBS VAL*,7X,*COMP VAL*,8X,*RESIDUAL*,//)
      SS=0.0
      TSS = 0.0
      TS=0.0
      DO 32 K=1,N
      COMP(K)=0.0
      RES(K)=0.0
      UA=(P*(U(K)-UO))/UW
      VA=(P*(V(K)-VO))/VW
      DO 33 I=1,M
      UII=II(I)*UA
      VII=JJ(I)*VA
      IF(IDT(I) .EQ. 1)37,34
   34 IF(IDT(I) .EQ. 2)38,35
   35 IF(IDT(I) .EQ. 3)39,36
   36 COMP(K)=COMP(K)+C(I)*SINF(UII)*SINF(VII)
      GO TO 40
   37 COMP(K)=COMP(K)+C(I)*COSF(UII)*COSF(VII)
      GO TO 40
   38 COMP(K)=COMP(K)+C(I)*COSF(UII)*SINF(VII)
      GO TO 40
   39 COMP(K)=COMP(K)+C(I)*SINF(UII)*COSF(VII)
   40 CONTINUE
   33 CONTINUE
      RES(K)=D(K)-COMP(K)
      PRINT 111, IDEN(K),U(K),V(K),D(K),COMP(K),RES(K)
  111 FORMAT(10X,I4,5X,2(F6.2,4X),3(F10.4,5X))
      SS=RES(K)*RES(K)+SS
      TS=TS+D(K)
   32 TSS=TSS+D(K)*D(K)
      TS=TS/N
      AS=TS*TS*N
      VAR=(TSS-AS)/(N-1)
      ST=VAR**0.5
      ZS=100.0*(1.0-(SS/(TSS-AS)))
      F=((N-M)*(TSS-AS-SS))/((M-1)*SS)
      IDS=M-1
      IDR=N-M
      PRINT 112
  112 FORMAT(///10X,*DATA MEAN*,8X,*ST DEVIATION*,7X,*VARIANCE*,7X,
     X*PCT SS CONTRIBN*,8X,*F RATIO*,2X,*DF NUM*,2X,*DF DENOM*)
      PRINT 113,TS,ST,VAR,ZS,F,IDS,IDR
```

```
  113 FORMAT(/10X,5(F12.4,5X),2(I3,5X))
   50 CONTINUE
C*************************************
C  MAPPING OPTION FOR COMPUTED SURFACE
C*************************************
      IF(OPB .EQ. 1.0)51,69
   51 IF(ITOP .EQ. 1)70,52
   52 DO 53 I=1,42
      UU(I)=((P*(UL-UO))/UW)+(P*(I-1)*(UH-UL))/(41.0*UW)
      DO 53 J=1,70
   53 CVAL(I,J)=0.0
      DO 54 J=1,70
   54 VV(J)=((P*(VL-VO))/VW)+(P*(J-1)*(VH-VL))/(69.0*VW)
      DO 63 K=1,42
      DO 63 L=1,70
      DO 55 I=1,M
      UII=II(I)*UU(K)
      VII=JJ(I)*VV(L)
      IF(IDT(I) .EQ. 1)58,56
   56 IF(IDT(I) .EQ. 2)59,57
   57 IF(IDT(I) .EQ. 3)60,61
   58 CVAL(K,L)=CVAL(K,L)+C(I)*COSF(UII)*COSF(VII)
      GO TO 62
   59 CVAL(K,L)=CVAL(K,L)+C(I)*COSF(UII)*SINF(VII)
      GO TO 62
   60 CVAL(K,L)=CVAL(K,L)+C(I)*SINF(UII)*COSF(VII)
      GO TO 62
   61 CVAL(K,L)=CVAL(K,L)+C(I)*SINF(UII)*SINF(VII)
   62 CONTINUE
   55 CONTINUE
   63 CONTINUE
      CALL COZYMAP(1)
      PRINT 4005
 4005 FORMAT(1H1,20X,*SYMBOL MEANINGS*,//10X,*SYMBOL*,5X,*MAX VAL*,//)
      DO 4000 K=1,19
      PRINT 4001, CH(K), CONTOUR(K)
 4001 FORMAT(13X,1A1,6X,F10.4)
 4000 CONTINUE
      GO TO 84
   70 DO 71 K=1,42
      UU(K)=((P*(UL-UO))/UW)+(P*(K-1)*(UH-UL))/(41.0*UW)
      DO 71 L=1,70
      S(K,L) =0.0
      VV(L)=((P*(VL-VO))/VW)+(P*(L-1)*(VH-VL))/(69.0*VW)
      DO 72 I=1,M
      UII=II(I)*UU(K)
      VII=JJ(I)*VV(L)
      IF(IDT(I) .EQ. 1)73,74
   74 IF(IDT(I) .EQ. 2)78,79
   79 IF(IDT(I) .EQ. 3)80,81
   73 S(K,L)=S(K,L)+C(I)*COSF(UII)*COSF(VII)
      GO TO 82
   78 S(K,L)=S(K,L)+C(I)*COSF(UII)*SINF(VII)
      GO TO 82
   80 S(K,L)=S(K,L)+C(I)*SINF(UII)*COSF(VII)
      GO TO 82
   81 S(K,L)=S(K,L)+C(I)*SINF(UII)*SINF(VII)
   82 CONTINUE
   72 CONTINUE
   71 CONTINUE
      CALL COZYMAP(2)
```

```
      DO 83 I=1,42
      DO 83 J=1,70
   83 S(I,J)=S(I,J)-CVAL(I,J)
      CALL COZYMAP(3)
      GO TO 1
   69 CONTINUE
   84 IF(OPC .EQ. 1.0)75,99
   75 READ(60,103) (II(K),JJ(K),IDT(K),K=1,N)
      M=N
      ITOP=1
      GO TO 76
   99 GO TO 1
      END
      SUBROUTINE COZYMAP(IOP)
      DIMENSION PR(140)
      COMMON/BB/ VV(70) ,UU(42),CH(20), BASE, CONTOUR(19), BLANK,
     X S(90,90), P,UW,VW,UL,UH,VL,VH,UO,VO, CVAL(42,70)
      DATA (CH=1H1,1H8,1H/,1H5,1H.,1H3,1H*,1H2,1H+,1H6,1H,,1H9,1H-,1H0,
     X1H=,1H4,1H(,1H7,1H),1H$)
      DATA(BLANK=1H )
      IF(IOP .EQ. 1)1,2
    1 PRINT 100
  100 FORMAT(1H1,//,20X,*TREND SURFACE*)
      GO TO 5
    2 IF(IOP .EQ. 2)3,4
    3 PRINT 101
  101 FORMAT(1H1,//20X,*TOTAL SURFACE, FOURIER FIT*)
      GO TO 5
    4 PRINT 102
  102 FORMAT(1H1,//,20X,*RESIDUALS FROM FOURIER TREND*)
    5 CONTINUE
      PRINT 150
  150 FORMAT(/19X,1HU,30X,1HV)
      DO 6 J=1,70
    6 VV(J)=VL + ((J-1)*(VH-VL))/69.0
      DO 7 I=1,42
    7 UU(I)=UL + ((I-1)*(UH-UL))/41.0
      PRINT 111, VV(1),(VV(J),J=10,70,10)
  111 FORMAT(//21X,F6.2,7(F6.2,4X),/24X,1H*,8X,1H*,6(9X,1H*))
      PRINT 112
  112 FORMAT(23X,72(1H-))
      IF(IOP .EQ. 1) 50,51
   50 DO 52 I=1,42
      DO 52 J=1,70
   52 S(I,J)=CVAL(I,J)
   51 CONTINUE
      DO 10 I=1,42
      DO 9 J=1,70
      PR(J)=BLANK
      IF(S(I,J) .LT. BASE) GO TO 14
      DO 11 K=1,19
      IF(S(I,J) .LE. CONTOUR(K)) GO TO 13
   11 CONTINUE
      K=20
   13 PR(J)=CH(K)
   14 CONTINUE
    9 CONTINUE
      IF(I .EQ. 1 .OR. MOD(I,6) .EQ. 0) GO TO 15
      PRINT 113, (PR(J),J=1,70)
  113 FORMAT(23X,1HI,70A1,1HI)
```

```
      GO TO 10
   15 PRINT 114,UU(I),(PR(J),J=1,70)
  114 FORMAT(16X,F6.2,2H*I,70A1,1HI)
   10 CONTINUE
      PRINT 112
      END
```

## EXAMPLES

Isopach data for an evaporite basin in Colorado and Kansas were taken from Krumbein (1962). There are 31 irregularly spaced data points. The location of these points on a U, V grid and the corresponding well numbers are shown in Figure 2. U, V values and observed thicknesses are shown in Table 2 below.
A hand-contoured map is presented in Krumbein (1966 p. 2236). These data provide the bases for the five maps shown in Figures 3-7.

Figure 3 is the first harmonic trend (Block 0 + I as indicated in Figure 1) of the above data with a fundamental wavelength chosen as six units in the U and V directions. Locations of control points are shown by heavy dots. Figure 4 is also the first harmonic trend surface but with fundamental wavelengths of 12 units in the U and V directions. The difference between these two maps is apparent although the sum of squares reduction is but slightly different. In Figure 3 the basin seemingly is symmetrical, whereas in Figure 4 the basin seems to have a somewhat linear trend to the northwest.

Table 2.- Isopach data for map example.

| Well no. | U | V | Thickness of unit in feet |
|---|---|---|---|
| 1001 | 3.55 | 3.10 | 845 |
| 1004 | 3.40 | 2.30 | 906 |
| 1006 | 3.30 | 1.15 | 844 |
| 1007 | 2.95 | 0.20 | 447 |
| 1009 | 4.85 | 3.10 | 1001 |
| 1010 | 5.00 | 2.60 | 933 |
| 1012 | 4.35 | 0.60 | 374 |
| 1014 | 2.60 | 1.85 | 608 |
| 1015 | 2.85 | 2.35 | 640 |
| 1017 | 4.30 | 1.15 | 614 |
| 1019 | 3.80 | 2.90 | 915 |
| 1020 | 4.00 | 3.60 | 1139 |
| 1021 | 4.95 | 2.25 | 702 |
| 1023 | 2.30 | 2.60 | 464 |
| 2002 | 3.65 | 3.70 | 1118 |
| 2003 | 4.20 | 3.85 | 1224 |
| 2004 | 4.40 | 4.25 | 1204 |
| 2005 | 5.10 | 4.10 | 1144 |
| 2006 | 5.50 | 3.80 | 1048 |
| 2008 | 3.45 | 4.80 | 1162 |
| 2009 | 3.30 | 5.10 | 1003 |
| 2011 | 3.10 | 5.55 | 721 |
| 2012 | 3.00 | 6.20 | 775 |
| 2015 | 5.50 | 4.20 | 1023 |
| 2016 | 5.30 | 4.30 | 1114 |
| 2017 | 4.60 | 5.70 | 955 |
| 2019 | 2.20 | 4.50 | 532 |
| 2021 | 2.30 | 5.50 | 562 |
| 2031 | 5.10 | 5.75 | 1005 |
| 2034 | 1.40 | 5.55 | 530 |
| 8001 | 5.80 | 3.40 | 1126 |

Figures 5 and 6 show these same surfaces on twice the scale of Figures 3 and 4 with extrapolation beyond the control area included. Figure 5 shows extrapolation of the first harmonic trend with a fundamental wavelength of six units in the U and V directions. It is seen that this surface is merely duplicated over every interval of six units. Figure 6 shows extrapolation of the trend with 12 units as a fundamental wavelength. There is no repetition of the surface in this map area as the extrapolation was only carried out over one fundamental wavelength. Strongly negative values, however, appear on the map both to the northwest and southwest of the control area. This map would begin to duplicate itself in the U and V directions if any further extrapolation was carried out.

Figure 7 shows the complete fit surface to the data using a fundamental wavelength of twelve units. The blank area indicates strongly negative values and the dollar symbols indicate strongly positive values. It is apparent that extrapolation of this total fit map is unreasonable.

## REFERENCES


Harbaugh, J. W., and Preston, F. W., 1965, Fourier series analysis in geology: Symposium on computer and computer applications in mining and exploration: Univ. Arizona, Tucson, v. 1, p. R1-R46.

James, W. R., 1966, The Fourier series in Map analysis, Tech. Report No. 1, ONR Contract Nonr-1228(36), Office of Naval Research, Geography Branch, 37 p.

Krumbein, W. C., 1962, Open and closed number systems in stratigraphic mapping: Am. Assoc. Petroleum Geologists Bull., v. 46, p. 2229-2245.

———, 1966, A comparison of polynomial and Fourier models in map analysis: Tech. Report No. 2, ONR Contact Nonr-1228(36), Office of Naval Research, Geography Branch, 45 p.

Preston, F. W., and Harbaugh, J. W., 1965, BALGOL program and geologic application for single and double Fourier series using IBM 7090/94 computers: Kansas Geol. Survey Sp. Dist. Publ. 24, 72 p.




Figure 2.- Location of wells on U,V grid.




Figure 3.- First harmonic trend with fundamental wavelength of six units in U and V directions.




Figure 4.- First harmonic trend with fundamental wavelength of twelve units in U and V directions.




Figure 5.- Extrapolated first harmonic trend with fundamental wavelength of six units.




Figure 6.- Extrapolated first harmonic trend with fundamental wavelength of twelve units.




Figure 7.- Total fit surface with fundamental wavelength of twelve units.

KANSAS GEOLOGICAL SURVEY COMPUTER PROGRAM
THE UNIVERSITY OF KANSAS, LAWRENCE

PROGRAM ABSTRACT

Title (If subroutine state in title):

FOURFIT (main program title)

SUBROUTINE COZYMAP

Computer: CDC 3400 Date: September 22, 1966

Programming language: FORTRAN IV (3400)

Author, organization: William R. James, Department of Geology, Northwestern University, Evanston, Illinois.

Direct inquiries to: Author, or

Name: D. F. Merriam Address: Kansas Geological Survey
University of Kansas, Lawrence

Purpose/description: A double Fourier series is used as a mapping function for use in trend-surface analysis on irregularly spaced data. Coefficients are computed for given sets of double Fourier series terms and continuous symbol maps are produced from them.

Mathematical method: Coefficients are derived by the method of least squares.

Restrictions, range: Up to 200 data points may be used and up to 90 coefficients may be calculated for any given map.

Storage requirements:

Equipment specifications: Memory 20K___ 40K___ 60K___ K 32

Automatic divide: Yes___ No___ Indirect addressing Yes X No___

Other special features required Floating point option.

Additional remarks (include at author's discretion: fixed/float, relocatability; optional: running time, approximate number of times run successfully, programming hours) Regardless of type of machine, in making a total fit with a matrix larger than 25 x 25, roundoff error becomes critical. This program was adapted to regular FORTRAN IV to run on the IBM 7040, and although the trend surfaces were usually similar in areas of good control, the coefficients were markedly different. Therefore, each investigator using this program will have to evaluate results in light of runs at his installation.

## COMPUTER CONTRIBUTIONS
Kansas Geological Survey
University of Kansas
Lawrence, Kansas

Computer Contribution

1. Mathematical simulation of marine sedimentation with IBM 7090/7094 computers, by J. W. Harbaugh, 1966 . . . $1.00
2. A generalized two-dimensional regression procedure, by J. R. Dempsey, 1966 . . . . . . . . . . $0.50
3. FORTRAN IV and MAP program for computation and plotting of trend surfaces for degrees 1 through 6, by Mont O'Leary, R. H. Lippert, and O. T. Spitz, 1966 . . . . . . . . . . . . . . . $0.75
4. FORTRAN II program for multivariate discriminant analysis using an IBM 1620 computer, by J. C. Davis and R. J. Sampson, 1966 . . . . . . . . . . . . . . . . . $0.50
5. FORTRAN IV program using double Fourier series for surface fitting of irregularly spaced data, by W. R. James, 1966. . . . . . . . . . . . . . . . . . $0.75

Special Distribution Publication

3. BALGOL program for trend-surface mapping using an IBM 7090 computer, by J. W. Harbaugh, 1963 . . . . . $0.50
4. FORTRAN II program for coefficient of association (Match-Coeff) using an IBM 1620 computer, by R. L. Kaesler, F. W. Preston, and D. I. Good, 1963 . . . . . . . . . . . . . . . $0.25
9. BALGOL programs for calculation of distance coefficients and correlation coefficients using an IBM 7090 computer, by J. W. Harbaugh, 1964 . . . . . . . . . . . . . . . $0.75
11. Trend-surface analysis of regional and residual components of geologic structure in Kansas, by D. F. Merriam and J. W. Harbaugh, 1964 . . . . . . . . . . . . . . . $0.75
12. FORTRAN and FAP program for calculating and plotting time-trend curves using an IBM 7090 or 7094/1401 computer system, by W. T. Fox, 1964 . . . . . . . . . . . . . . . $0.75
13. FORTRAN program for factor and vector analysis of geologic data using an IBM 7090 or 7094/1401 computer system, by Vincent Manson and John Imbrie, 1964 . . . . . . . . . . . . . . . $1.00
14. FORTRAN II trend-surface program for the IBM 1620, by D. I. Good, 1964 . . . . . . . . . . . $1.00
15. Application of factor analysis to petrologic variations of Americus Limestone (Lower Permian), Kansas and Oklahoma, by J. W. Harbaugh and Ferruh Demirmen, 1964 . . . . . . . . . . . . . . . $1.00
23. ALGOL program for cross-association of nonnumeric sequences using a medium-size computer, by M. J. Sackin, P.H.A. Sneath, and D. F. Merriam, 1965 . . . . . . . . . . . . . . . $0.75
24. BALGOL program and geologic application for single and double Fourier series using IBM 7090/7094 computers, by F. W. Preston and J. W. Harbaugh, 1965 . . . . . . . . . . . . . . . $1.00
25. Final report of the Kansas Geological Society Basement Rock Committee and list of Kansas wells drilled into Precambrian rocks, by V. B. Cole, D. F. Merriam, and W. W. Hambleton, 1965 . . . . . . . . . $0.75
26. FORTRAN II trend-surface program with unrestricted input for the IBM 1620 computer, by R. J. Sampson and J. C. Davis, 1966. . . . . . . . . . . . . . . . . $0.50
27. Application of factor analysis to a facies study of the Leavenworth Limestone (Pennsylvanian-Virgilian) of Kansas and environs, by D. F. Toomey, 1966 . . . . . . . . . . . . . . . $0.75
28. FORTRAN II program for standard-size analysis of unconsolidated sediments, by J. W. Pierce and D. I. Good, 1966 $0.75

Reprints (available for limited time)

Geology and the computer, by D. F. Merriam (reprinted from New Scientist, v. 26, no. 444, 1965). . . . . . no charge

Quantitative comparison of contour maps, by D. F. Merriam and P. H. A. Sneath (reprinted from Journal of Geophysical Research, v. 71, no. 4, 1966) . . . . . . . . . . . . . . . no charge

Trend-surface analysis of stratigraphic thickness data from some Namurian rocks east of Sterling, Scotland, by W. A. Read and D. F. Merriam (reprinted from Scottish Journal of Geology, v. 2, pt. 1, 1966) . . . . . . . no charge

Generation of orthogonal polynomials for trend surfacing with a digital computer, by O. T. Spitz (reprinted from Computers and operations research in mineral industries, Pennsylvania State University, 6th Annual Symposium, 1966) . . . . . . . . . . . . . . . . . no charge

The use of statistical communication theory for characterization of porous media, by F. W. Preston, D. W. Green, and W. D. Aldenderfer (reprinted from Computers and operations research in mineral industries, Pennsylvania State University, 6th Annual Symposium, 1966) . . . . . . . . . . . . . . . no charge

Geologic model studies using trend-surface analysis, by D. F. Merriam and R. H. Lippert (reprinted from Journal of Geology, v. 74, no. 5, 1966) . . . . . . . . . . . . . . . no charge

Geologic use of the computer, by D. F. Merriam (reprinted from Wyoming Geological Association, 20th Field Conference, 1966 . . . . . . . . . . . . . . . . . no charge